// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/security/Pausable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (security/Pausable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which allows children to implement an emergency stop
 * mechanism that can be triggered by an authorized account.
 *
 * This module is used through inheritance. It will make available the
 * modifiers `whenNotPaused` and `whenPaused`, which can be applied to
 * the functions of your contract. Note that they will not be pausable by
 * simply including this module, only once the modifiers are put in place.
 */
abstract contract Pausable is Context {
    /**
     * @dev Emitted when the pause is triggered by `account`.
     */
    event Paused(address account);

    /**
     * @dev Emitted when the pause is lifted by `account`.
     */
    event Unpaused(address account);

    bool private _paused;

    /**
     * @dev Initializes the contract in unpaused state.
     */
    constructor() {
        _paused = false;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is not paused.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    modifier whenNotPaused() {
        _requireNotPaused();
        _;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is paused.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    modifier whenPaused() {
        _requirePaused();
        _;
    }

    /**
     * @dev Returns true if the contract is paused, and false otherwise.
     */
    function paused() public view virtual returns (bool) {
        return _paused;
    }

    /**
     * @dev Throws if the contract is paused.
     */
    function _requireNotPaused() internal view virtual {
        require(!paused(), "Pausable: paused");
    }

    /**
     * @dev Throws if the contract is not paused.
     */
    function _requirePaused() internal view virtual {
        require(paused(), "Pausable: not paused");
    }

    /**
     * @dev Triggers stopped state.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    function _pause() internal virtual whenNotPaused {
        _paused = true;
        emit Paused(_msgSender());
    }

    /**
     * @dev Returns to normal state.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    function _unpause() internal virtual whenPaused {
        _paused = false;
        emit Unpaused(_msgSender());
    }
}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be _NOT_ENTERED
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Returns true if the reentrancy guard is currently set to "entered", which indicates there is a
     * `nonReentrant` function in the call stack.
     */
    function _reentrancyGuardEntered() internal view returns (bool) {
        return _status == _ENTERED;
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.4) (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }

    function _contextSuffixLength() internal view virtual returns (uint256) {
        return 0;
    }
}

// File: contracts/IUniFiReferral.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.16;

interface IUniFiReferral {
    // Events
    event RefCodeCreated(uint256 refCodeId, address indexed owner, bytes5 refCodeHash);
    event RefCodeApplied(uint256 refCodeId, address indexed applier, bytes5 refCodeHash);
    
    // Functions
    function ethRequired() external view returns (uint256);
    function uniFiRequired() external view returns (uint256);
    function treasuryAddress() external view returns (address);
    function UniFiAddress() external view returns (address);

    function createRefETH(bytes5 _refCodeHash) external payable;
    function createRefUniFi(bytes5 _refCodeHash, uint256 _value) external payable;
    function applyRefCode(bytes5 _refCodeHash) external;
    function presaleRef(bytes5 _refCodeHash, address _owner) external;
    function isRefCodeTaken(bytes5 _refCodeHash) external view returns (bool);
    function getReferrer(address _addr) external view returns (address);
    function getOwnedRefCodes(address _addr, uint256 start, uint256 end) external view returns (bytes5[] memory);
    function getAppliedRefCodeHash(address _addr) external view returns (bytes5);
    function getAppliedRefCodeId(address _addr) external view returns (uint256);
    function getRefCodeHash(uint256 _refCodeId) external view returns (bytes5);
    function getRefCodeId(bytes5 _refCodeHash) external view returns (uint256);
    function getReferralPoints(uint256 _refCodeId) external view returns (uint256);
}

// File: contracts/UniFiPresale.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.16;

import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "@openzeppelin/contracts/access/Ownable.sol";
import "./IUniFiReferral.sol";
import "@openzeppelin/contracts/security/Pausable.sol";
import "@openzeppelin/contracts/security/ReentrancyGuard.sol";

contract UniFiPresale is Ownable, Pausable, ReentrancyGuard {
    address public tokenAddress = 0xb131f4A55907B10d1F0A50d8ab8FA09EC342cd74;
    address public referralAddress;
    address public UniFiAddress;

    uint256 public whitelistStartTime = 1705561200;
    uint256 public publicSaleStartTime = 1705647600;
    uint256 public presaleEndTime = 1705820400;
    uint256 public minEthAmount = 0.001 ether;
    uint256 public maxEthAmount = 1 ether;
    uint256 public totalTokensForSale = 200000 * 10**5; // 197,280 tokens by default
    uint256 public individualHoldings = 69 * 10**18;
    uint256 public claimStartTime = 1705993200;
    uint256 public tokenPrice = 0.001 ether;
    uint256 public whitelistDiscount = 0.0001 ether;

    mapping(address => uint256) public allocations;
    mapping(address => uint256) public commits;

    event Allocation(address indexed buyer, uint256 amountEth, uint256 tokens);
    event Claimed(address indexed buyer, uint256 tokens);

    // View whitelist status
    function isWhitelisted(address _user) public view returns (bool) {
        return IERC20(tokenAddress).balanceOf(_user) >= individualHoldings;
    }

    // Set token address
    function setTokenAddress(address _newToken) external onlyOwner {
        tokenAddress = _newToken;
    }

    // Set UniFi address
    function setUniFi(address _newUniFi) external onlyOwner {
        UniFiAddress = _newUniFi;
    }

    // Set referral address
    function setReferral(address _newReferral) external onlyOwner {
        referralAddress = _newReferral;
    }

    // Set the individual token holdings required for whitelisting
    function setIndividualHoldings(uint256 _newHoldings) external onlyOwner {
        individualHoldings = _newHoldings;
    }

    // Set the minimum and maximum ETH amounts for each purchase
    function setCaps(uint256 _minEthAmount, uint256 _maxEthAmount)
        external
        onlyOwner
    {
        minEthAmount = _minEthAmount;
        maxEthAmount = _maxEthAmount;
    }

    // Set the total number of tokens available for sale
    function setTotalTokensForSale(uint256 _newTotal) external onlyOwner {
        totalTokensForSale = _newTotal;
    }

    // Set the timestamp when claiming becomes available
    function setClaimStartTime(uint256 _newClaimStartTime) external onlyOwner {
        claimStartTime = _newClaimStartTime;
    }

        // Set whitelist start time
    function setWhitelistStartTime(uint256 _newStartTime) external onlyOwner {
        whitelistStartTime = _newStartTime;
    }

    // Set public sale start time
    function setPublicSaleStartTime(uint256 _newStartTime) external onlyOwner {
        publicSaleStartTime = _newStartTime;
    }

    // Set presale end time
    function setPresaleEndTime(uint256 _newEndTime) external onlyOwner {
        presaleEndTime = _newEndTime;
    }

    // Set the token price in ETH
    function setTokenPrice(uint256 _newTokenPrice) external onlyOwner {
        tokenPrice = _newTokenPrice;
    }

    // Set the whitelist discount in ETH
    function setDiscount(uint256 _newDiscount) external onlyOwner {
        whitelistDiscount = _newDiscount;
    }

    // Function to handle the presale purchase
    function buyPresale() external payable whenNotPaused nonReentrant {
        // Check if the presale is active
        require(
            block.timestamp >= whitelistStartTime &&
                block.timestamp < presaleEndTime,
            "Presale is not active"
        );

        uint256 buyingPrice = tokenPrice;
        // If within the whitelisting period, check the buyer's balance
        if (
            block.timestamp >= whitelistStartTime &&
            block.timestamp < publicSaleStartTime
        ) {
            require(
                IERC20(tokenAddress).balanceOf(msg.sender) >= individualHoldings,
                "Not whitelisted"
            );
            buyingPrice = tokenPrice - whitelistDiscount;
        }

        // Check if the sent ETH amount is within the allowed range
        require(
            msg.value >= minEthAmount && msg.value <= maxEthAmount,
            "Invalid ETH amount"
        );

        // Calculate the token amount based on the ETH sent
        uint256 allocationAmount = msg.value;
        uint256 tokensToBuy = calculateTokens(allocationAmount, buyingPrice);

        // Ensure the user doesn't buy more tokens than available
        if (tokensToBuy >= totalTokensForSale) {
            tokensToBuy = totalTokensForSale;
        }

        // Update user's allocation and reduce total tokens for sale
        allocations[msg.sender] += tokensToBuy;
        commits[msg.sender] += allocationAmount;
        totalTokensForSale -= tokensToBuy;

        // Emit Allocation event
        emit Allocation(msg.sender, allocationAmount, tokensToBuy);
    }

    // Function to calculate token amount based on ETH sent
    function calculateTokens(uint256 _ethAmount, uint256 _buyingPrice) public pure returns (uint256) {
        return (_ethAmount * 10**5) / _buyingPrice;
    }

    // Function to claim purchased tokens
    function claimPresale(bytes5 _refCode) external whenNotPaused nonReentrant {
        // Check if claiming is allowed
        require(
            block.timestamp >= claimStartTime,
            "Claiming is not allowed yet"
        );

        // Check if the user has an allocation
        require(allocations[msg.sender] > 0, "No allocation");

        // Transfer tokens from contract to the buyer
        IERC20(UniFiAddress).transfer(
            msg.sender,
            allocations[msg.sender]
        );

        // Create referral
        IUniFiReferral(referralAddress).presaleRef(_refCode, msg.sender);

        // Update claimed amount
        emit Claimed(msg.sender, allocations[msg.sender]);
        allocations[msg.sender] = 0;
        commits[msg.sender] = 0;
    }

    // Fallback function to receive ETH
    receive() external payable {}

    // Function to withdraw remaining tokens after the presale ends
    function withdrawRemainingTokens() external onlyOwner {
        IERC20(UniFiAddress).transfer(owner(), totalTokensForSale);
    }

    // Function to withdraw remaining ETH after the presale ends
    function withdrawRemainingETH() external onlyOwner {
        payable(owner()).transfer(address(this).balance);
    }

    function pause() external onlyOwner {
        _pause();
    }

    function unpause() external onlyOwner {
        _unpause();
    }

    function refund() external whenPaused nonReentrant {
        require(commits[msg.sender] > 0, "No allocations");
        
        uint256 refundAmount = commits[msg.sender];
        commits[msg.sender] = 0;
        allocations[msg.sender] = 0;
        
        payable(msg.sender).transfer(refundAmount);
    }
}